LIS Self Teaching Package
2019

SPSS version

Part II: Gender, employment, and wages
Part II
Gender, employment, and wages

Overall Plan and Structure of the Exercise

The exercises in Part I demonstrated the use of household income data along with useful programming techniques for working with the LIS data. Part II emphasises the use of person-level data, including wages, demographics, and labour market information. Whereas Part I consisted entirely of calculating descriptive statistics, Part II introduces regression modelling in the final exercises.

The program that was written in the first set of exercises is now completed and can be set aside. Starting with the next exercise, you will begin the process of building up an entirely new program for Part II. Many of the techniques shown in the previous part will be useful again. In addition, users will learn how to combine LIS datasets by merging household and person files, and by concatenating multiple country-year datasets into a single file.

The general approach of the exercises is the same as in Part I. After beginning a new program in the initial exercise, each subsequent exercise will add new code to the existing program. Within each exercise, results will be produced that help to illuminate the central research themes of this section.

Research Questions

The analysis of poverty and inequality using household income, which was covered in Part 1, has always been central to research using LIS data. Over the years, however, there has been an increasing volume of work that examines individual outcomes in the labor market. The richness of the labor market data available in LIS has increased over time, and today it is possible to address many types of questions about wages and employment.

Labor market outcomes for women are one especially popular area of research. Women's rate and intensity of work shows much wider cross-country variation than men's. At the same time, on average women consistently earn lower wages.
One reflection of this growing body of research is the recent creation of the LIS Employment Key Figures by Gender (http://www.lisdatacenter.org/lis-ekf-webapp/app/search-ekf-tables), which provide information about demographics, employment, earnings, and inequality by both country and gender. In the following exercises, we will investigate several of the topics covered by the key figures, although our results will differ somewhat due to sample selection and variable coding made for the exercises.

For the exercises, we will examine three countries, using data from LIS Wave VI: the United States, Germany, and Greece. As we will see, labor market outcomes for women show distinctly different patterns in each of these countries. Looking at persons of prime working age (which we will define as ages 25-54), our central questions will be:

- How does the percentage of prime-aged women employed in paid work vary across these three countries?
- Among those who are employed, how does the rate of part-time employment among women vary across the countries?
- How does employment vary by partnership and family status?
- How do wage differentials between men and women vary between across countries, across levels of educational attainment, and between immigrants and non-immigrants?

In the exercises, we will begin by producing tabulations of employment and wages for various population subgroups. In the concluding exercises, we will use linear regression to study multiple determinants of wages simultaneously, in order to better understand how family structure, education, and immigrant status are related to wages for men and for women.

**Before you begin**

Before you begin the exercises, take a look at the 2019 Template LIS User Guide, which can be accessed through LIS Website → Our Data → LIS Database. The User Guide provides an overview of the structure of LIS data and some data management practices, such as missing values policy and aggregation rules, which will be useful for working with LISSY.

In addition to this, an overview of the datasets and variables is provided through the METadata Information System (METIS) without having to login to LISSY. You can access METIS via the LIS Website → METIS → Enter
METIS → LIS. After selecting the datasets and variables, consult the Results tab for information on variables and definitions, dataset-specific information and variable availability across datasets.
Contents

1. Merging person and household data, selecting a sample
   - Selecting a sample of prime-age workers
   - Homeownership rates by country

2. Stacking data, employment rates by gender
   - Employment rates by country and gender
   - Part-time employment rates by country and gender

3. Family structure and employment
   - Family/partnership status and employment rates

4. Dependent employment and hourly wages
   - Non-dependent employment as a proportion of employment, by country
   - Gender wage gaps among dependent employees, by country

5. Hourly wages, education, and country-specific variables
   - Harmonised vs. country-specific codings
   - Gender wage gaps by country and educational attainment

6. Immigration and wages, understanding harmonisation
   - Understanding harmonization choices
   - Gender wage gaps by country and immigrant status

7. Wage regressions
   - Regressing log wages on demographics, separately by gender and country

8. Pooled regressions and normalised weights
   - PPP adjustments for income
   - Pooled regression with multiple countries, using normalised weights
1. Merging person and household data, selecting a sample

**Goal**

While the exercises in Part I used only data at the household level, Part II uses data from both the household and person level files. In this exercise, we will combine the person and household files in order to create a single dataset for each country in which household data is appended to each person record.

This exercise also selects the universe of persons that we will be studying in the subsequent exercises. Since we are interested in labor market outcomes, we will restrict our attention to people of *prime age*: those who are likely to be old enough to have completed schooling and young enough to not yet be retired. In these exercises, we will define prime age persons as those between 25 and 54 years old. This is a commonly used range in statistics from the United States government and other sources, but other definitions are also possible.

Some of the variables we will be using are not always available for household members other than the head and spouse. For that reason, we will further restrict our universe to heads and partners only.

All of the variables that will be needed in the subsequent exercises are introduced here. However, for now we will only analyse one: we will summarize home ownership, which is a household-level variable, in order to measure the rate of homeownership in each of the three countries under analysis. Homeownership will be included in our later multivariate analysis, because it serves as a rough proxy for wealth, which we otherwise have no information about.

**Activity**

Go to *LIS website → Login LISSY* tab and launch the Job Submission Interface (JSI) application. Open the downloaded application and logon to it with your LISSY account. Note that you may be prompted to update your Java application, which in most cases will not be required in order to launch LISSY.

Write a program to loop through three datasets: United States 2004 (*us04*), Germany 2004 (*de04*) and Greece 2004 (*gr04*). In each country, merge the person file to the household file and keep the following variables:

- Household: Unique household identifier (*hid*) and owned/rented housing indicator (*own*)
Person: Unique household identifier (hid), dataset name (dname), normalized person weight (pwgt), inflated person weight (ppopwgt), relationship to the household head (relation), partnership status (partner), age of youngest own child living in the household (ageyoch) age (age), sex (sex), immigrant indicator (immigr), 3-category recoded educational attainment (educ), country-specific educational attainment (educ_c), indicator for employment (emp), status in employment (status1), indicator for part-time employment (ptime1), and gross hourly wage in the first job (gross1).

Keep only those cases that are in the prime age range (between 25 and 54), and which are defined as either household heads or spouses in the variable relation.

Create an indicator variable equal to 1 if a person owns their house (with or without a mortgage), and 0 otherwise. Summarize this new variable to find the homeownership rate among the prime-aged persons for each country, and complete the following table:

<table>
<thead>
<tr>
<th>Dataset</th>
<th>Homeownership %</th>
</tr>
</thead>
<tbody>
<tr>
<td>US04</td>
<td></td>
</tr>
<tr>
<td>DE04</td>
<td></td>
</tr>
<tr>
<td>GR04</td>
<td></td>
</tr>
</tbody>
</table>

**Question:** In which country do the largest percentage of persons of prime working age own their houses, and in which country are homeownership rates lowest?

**Question:** Are all the non-homeowners renters? If not, what other housing types are possible?

**Guidelines**

- Merging two files can be done with the `match files` command of SPSS. As the number of observations differs between the person and household files, we need to perform a many-to-one match. Each case in the “table” will get its information copied one or more records in the “file”. The match can only take place based on a variable that is present in both files; the ‘by’ variable:

  ```
  Match files file = <personfile>
  /table = <householdfile>
  /by <variable>
  ```
In LIS datasets, the variable \texttt{hid} uniquely identifies households, and can be found in both the person and household files.

- Remember that when recoding variables, you can find a listing of the possible values of the original variable in \texttt{METIS}. In this case, go to the LIS Database information (\texttt{Enter METIS → LIS}). Select DE04, GR04 and US04. Select the variable \texttt{own}. Go to Results $→$ Crossed-compare and click on the variable name to see the statistics and labels of the variables.

- Like the household file, the person file contains weight variables. These variables can be used to weight by person, as an alternative to the method of multiplying household weights by number of household members that was used in the Part I. Although home ownership is a household-level variable, you will want to use the person weight to determine the proportion of \textit{persons} who live in owner-occupied dwellings. For now, use the variable \texttt{ppopwgt}, which inflates to the total population size.

- Please keep in mind that SPSS is sensitive to special characters. If you experience issues when copy-pasting the code from the self-teaching packages, you can first paste it to a Notepad file and then to the LISSY interface.
Program

define ex21 ()
select if age ge 25 and age le 54.
select if relation ge 1000 and relation le 2200.
if own ge 100 and own le 199 owner = 1.
if own ge 200 and own le 299 owner = 0.
weight by ppopwgt.
frequencies variables = owner.
!enddefine.

match files file = us04p
    /table = us04h
    /keep=did hid own dname pwgt ppopwgt relation partner ageyoch age sex immigr educ educ_c emp status1 ptime1 gross1
    /by hid.
ex21.

match files file = de04p
    /table = de04h
    /keep=did hid own dname pwgt ppopwgt relation partner ageyoch age sex immigr educ educ_c emp status1 ptime1 gross1
    /by hid.
ex21.

match files file = gr04p
    /table = gr04h
    /keep=did hid own dname pwgt ppopwgt relation partner ageyoch age sex immigr educ educ_c emp status1 ptime1 gross1
    /by hid.
ex21.
Results

<table>
<thead>
<tr>
<th>Dataset</th>
<th>Homeownership %</th>
</tr>
</thead>
<tbody>
<tr>
<td>US04</td>
<td>71.5%</td>
</tr>
<tr>
<td>DE04</td>
<td>42.7%</td>
</tr>
<tr>
<td>GR04</td>
<td>67.6%</td>
</tr>
</tbody>
</table>

**Question:** In which country do the largest percentage of persons of prime working age own their houses, and in which country are homeownership rates lowest?

- Homeownership rates are highest in the United States at 71.5%, and lowest in Germany at only 42.7%.

**Question:** Are all the non-homeowners renters? If not, what other housing types are possible?

- No, in these three datasets there is also a category labeled “free housing”. The LIS variable definitions available through [METIS](https://metis.econ.upenn.edu) explain that this can include “housing provided by employer, government or others, or illegal occupation”.

**Comments**

- You will notice that in this exercise the merge worked perfectly, i.e. all observations of the merging file were uniquely linked to one observation in the using file. This is always the case with LIS household and individual level files from the same dataset because all individuals belong to at least one and no more than one household.
2. Stacking data, employment rates by gender

Goal

So far, we have performed all the analysis separately for each dataset, working with only one country at a time. For this and all subsequent analyses, however, we will create a “stacked” dataset that contains information for all three countries in a single file. This means your dataset will have as many value observations as your countries altogether have stored within one file. This is necessary if you wish to analyse several countries simultaneously, which we will be doing in the regressions in exercise 2.8. In case of independent cross country comparisons, the use of a stacked dataset is an alternative technique to the looping through datasets.

After creating a combined dataset, we will examine rates of employment and part-time employment by gender, and see how it differs among these three countries. As in the previous exercise, we will be looking only at prime-aged persons who are defined as household heads or partners of household heads.

We will be using the LIS variable emp, an indicator that reports whether or not a person is currently employed. This variable will contain the current main employment status (as derived from LIS variable lfs). If the current main employment status is not available in the original dataset, employment status during income reference time or employment status according to the ILO criteria in the current period will be used in LIS variable lfs and hence also emp (you can check dataset-specific notes in METIS for information about each dataset). By this definition, a person may be considered as employed as soon as he/she has carried out any work.

Rates of employment and full-time employment among prime-age men tend to be similar and consistently high across countries, so we will be paying particular attention to differences in employment outcomes among women.

Activity

Modify your program so that it first creates a combined data file for the United States, Germany, and Greece, and then performs any necessary recoding and produces descriptive statistics.

Create a set of cross-tabulations that shows the rates of prime-age employment by gender within each country. Create another set of cross-tabulations showing the rates of part-time work by gender within each
country, among those who are employed. Use your results to complete the following table below.

<table>
<thead>
<tr>
<th>Dataset</th>
<th>Female employment rate</th>
<th>Part-time employment rate among employed women</th>
</tr>
</thead>
<tbody>
<tr>
<td>US04</td>
<td></td>
<td></td>
</tr>
<tr>
<td>DE04</td>
<td></td>
<td></td>
</tr>
<tr>
<td>GR04</td>
<td></td>
<td></td>
</tr>
</tbody>
</table>

**Question:** Contrast these countries in terms of their rates of female employment (high or low) and their rates of part-time employment among employed women (high or low).

**Guidelines**

- You do not need to remove the code that you used to produce the descriptive in the last exercise (on homeownership), but you can comment it out to make your job run slightly faster. To comment out a single line, place an asterisk (*) at the beginning of the line.

- When stacking datasets, you will need to save some temporary files. When saving temporary files, you can place them in a directory at LIS that is called by the macro-variable `mydata`. In order to ensure that your filename differs from others saving files at the same time, your filename should have a unique name. For example, you could include your LIS username in the filename:

  `save outfile = "mydata\exercisepart2_<myusername>.sav"`

- To put all of your country files together, you will need to append the datasets. Here is example code that will perform the person-to-household merge shown in the previous exercise, and then stack the three merged datasets:

  ```
  match files file = us04p
  /table = us04h
  /keep=did hid own dname pwgt ppopwgt relation partner ageyoch age sex immigr educ educ_c emp status1 ptime1 gross1
  /by hid.
  save outfile = "mydata\us04_<myusername>.sav".
  match files file = de04p
  /table = de04h
  /keep=did hid own dname pwgt ppopwgt relation partner ageyoch age sex immigr educ educ_c emp status1 ptime1 gross1
  ```
/by hid.
save outfile = "mydata\de04_<myusername>.sav".
match files file = gr04p 
/table = gr04h
/keep=did hid own dname pwgt ppopwgt relation partner ageyoch age sex immigr educ educ_c emp status1 ptime1 gross1
/by hid.
save outfile = "mydata\gr04_<myusername>.sav".
add files file = "mydata\us04_<myusername>.sav" 
/file = "mydata\de04_<myusername>.sav"
/file = "mydata\gr04_<myusername>.sav".
save outfile = "mydata\stackex_<myusername>.sav".

Once the stacked file is created, it will remain on the server, so you will be able to comment out the file-creation code in the next exercises.

- Once you have produced the stacked data file, you can produce descriptive for multiple countries at once by using the dname variable to split the file into parts and perform the same command repeatedly for subsets of the data. For example:
  
  Sort cases by dname.
  Split file by dname.
  frequencies variables = owner.

  This code will produce the summary statistics on homeownership by country that we produced in the previous exercise.

- You can get the row percentages in a cross-tabulation (such as gender by employment) by using the option /cells = row on the command:
  
  crosstabs sex by emp /cells = row.
Program

define ex22 ().

select if age ge 25 and age le 54.
select if relation ge 1000 and relation le 2200.
if own ge 100 and own le 199 owner = 1.
if own ge 200 and own le 299 owner = 0.
weight by ppopwgt.
sort cases by dname.
split file by dname.
* frequencies variables = owner.
crosstabs sex by emp /cells = row.
select if emp eq 1.
crosstabs sex by ptime1 /cells = row.
!enddefine .

match files file = us04p
/table = us04h
 /keep=did hid own dname pwgt ppopwgt relation partner ageyoch age sex immigr educ educ_c emp status1 ptime1 gross1
 /by hid.
save outfile = "mydata\us04_yid.sav".
match files file = de04p
/table = de04h
 /keep=did hid own dname pwgt ppopwgt relation partner ageyoch age sex immigr educ educ_c emp status1 ptime1 gross1
 /by hid.
save outfile = "mydata\de04_yid.sav".
match files file = gr04p
/table = gr04h
 /keep=did hid own dname pwgt ppopwgt relation partner ageyoch age sex immigr educ educ_c emp status1 ptime1 gross1
 /by hid.
save outfile = "mydata\gr04_yid.sav".
add file file = "mydata\us04_yid.sav"
/file = "mydata\de04_yid.sav"
/file = "mydata\gr04_yid.sav".
save outfile = "mydata\stackex_yid.sav".
ex22.
Results

<table>
<thead>
<tr>
<th>Dataset</th>
<th>Female employment rate</th>
<th>Part-time employment rate among employed women</th>
</tr>
</thead>
<tbody>
<tr>
<td>US04</td>
<td>72.0%</td>
<td>17.7%</td>
</tr>
<tr>
<td>DE04</td>
<td>72.2%</td>
<td>42.6%</td>
</tr>
<tr>
<td>GR04</td>
<td>57.6%</td>
<td>21.3%</td>
</tr>
</tbody>
</table>

Question: Contrast these countries in terms of their rates of female employment (high or low) and their rates of part-time employment among employed women (high or low).

- Employment rates among prime age workers are relatively high in the United States in Germany, and lower in Greece. In the United States, most employed women work full time, while nearly half of employed German women work part time. Greece combines low employment rates with high rates of full time employment among those women who are employed.
3. Family structure and employment

**Goal**

In the previous exercise we examined cross-national differences in women's employment. In this exercise, we will examine the variation in employment rates among women, based on their partnership and family status. We will contrast partnered and single women. Within each of those two categories, we will contrast women without children in the household, women with young children, and women with older children. The variables created in this exercise will be useful later, when we combine family structure with other personal characteristics in a multivariate analysis of wages.

**Activity**

Since you already created the merged, stacked dataset in the previous exercise, you do not need to create it again. Modify your code so that the subroutine that merges and stacks the data is commented out, and add a line that simply loads the merged and stacked file at the beginning of the program.

Create a variable `achildcat`, to indicate the age of the youngest own child living in the household. This variable should be equal to 0 if there are no children under 18, equal to 1 if the youngest child is under 6 years old, and equal to 2 if the youngest child is between 6 and 17. You can create this variable based on the information in the variable `children`.

Using the new variable you have created (`achildcat`), an indicator of whether a person is currently living with a partner (=1) or not (=0) (`partner`) and the employment indicator you used in the previous exercise (`emp`), produce summary statistics to fill the table below.

**Employment Rates**

| Dataset | All women | Partnered | | Single | | |
|---------|-----------|-----------|---|---|
|         |           | No children under 18 | Child under 6 | Child 6-17 | No children under 18 | Child under 6 | Child 6-17 |
| US04    |           |           |              |            |              |              |
| DE04    |           |           |              |            |              |              |
| GR04    |           |           |              |            |              |              |

**Question:** Within each country, which subpopulation of prime age women has the lowest employment rates?
**Guidelines**

- You can generate a new variable based on the values of an old one using the `recode` command:

  ```
  recode ageyoch (0 thru 5=1) (6 thru 17=2) (ELSE=0) into achildcat.
  ```

- You may find it helpful to attach labels to your new variables using the `add variable labels` and `add value labels` commands, as in the following example:

  ```
  add variable label achildcat "<labelname>".
  add value labels achildcat
  0 "no children under 18"
  1 "under 6 years"
  2 "6-17 years".
  ```

- As in the last exercise, you can use the `split file` command to calculate summary statistics over subsets of the data. It is possible to use this command with more than one variable, which will allow you to analyze subsets within subsets. For example:

  ```
  sort cases by dname sex partner.
  split file by dname sex partner.
  crosstabs achildcat by emp /cells = row.
  ```

This code will produce a tabulation of employment which is separated by country, then by gender, and then by partnership status, with row percentages.
**Program**

define ex23 ().
select if age ge 25 and age le 54.
select if relation ge 1000 and relation le 2200.
select if sex eq 2.
weight by ppopwgt.
recode ageyoch (0 thru 5=1) (6 thru 17=2) (ELSE=0) into achildcat.
add value labels achildcat
0 "no children under 18"
1 "under 6 years"
2 "6-17 years".
sort cases by dname partner.
split file by dname partner.
crosstabs achildcat by emp /cells = row.
!enddefine .

*match files file = us04p
 /table = us04h
 /keep=did hid own dname pwgt ppopwgt relation partner ageyoch age sex immigr educ educ_c emp status1 ptime1 gross1
 /by hid.
*save outfile = "mydata\us04_yid.sav".

*match files file = de04p
 /table = de04h
 /keep=did hid own dname pwgt ppopwgt relation partner ageyoch age sex immigr educ educ_c emp status1 ptime1 gross1
 /by hid.
*save outfile = "mydata\de04_yid.sav".

*match files file = gr04p
 /table = gr04h
 /keep=did hid own dname pwgt ppopwgt relation partner ageyoch age sex immigr educ educ_c emp status1 ptime1 gross1
 /by hid.
*save outfile = "mydata\gr04_yid.sav".
*add file file = "mydata\us04_yid.sav"
 /file = "mydata\de04_yid.sav"
 /file = "mydata\gr04_yid.sav".
*save outfile = "mydata\stackex_yid.sav".
get file = "mydata\stackex_yid.sav".
ex23.
**Results**

**Employment Rates**

<table>
<thead>
<tr>
<th>Dataset</th>
<th>All women</th>
<th>Partnered</th>
<th>Single</th>
</tr>
</thead>
<tbody>
<tr>
<td></td>
<td>No children under 18</td>
<td>Child under 6</td>
<td>Child 6-17</td>
</tr>
<tr>
<td>US04</td>
<td>72.0%</td>
<td>76.2%</td>
<td>58.4%</td>
</tr>
<tr>
<td>DE04</td>
<td>72.2%</td>
<td>80.8%</td>
<td>50.8%</td>
</tr>
<tr>
<td>GR04</td>
<td>57.6%</td>
<td>49.8%</td>
<td>57.7%</td>
</tr>
</tbody>
</table>

**Question:** Within each country, which subpopulation of prime age women has the lowest employment rates?

- In the United States, partnered women with young children have the lowest employment rates. In Greece, partnered women without children have the lowest employment rates. In Germany, however, single mothers with young children have lower employment rates. This may be because of more generous child policy in Germany that makes it easier for mothers of young children to support themselves without paid employment.

**Comments**

- There is no clear-cut definition of a single-mother household. In this exercise, we allow other adult members to be present (as long as they are not defined as her partner). An alternative approach would be to limit the sample to households composed of a single female adult and her children. Another possibility is to limit single mother households to those with children under a specified age limit.

- When subdividing subsets of the population as has been done here, pay attention to sample sizes. In small datasets, estimates for narrowly defined groups may become very small, making estimates less reliable. The estimate for single Greek women with young children in this exercise, for example, is based on only 17 cases!
4. Dependent employment and hourly wages

**Goal**

In the next several exercises, we will shift from considering employment to analysing the earnings of those who are employed. We will focus our analysis on the hourly wages and thus restrict our sample to those in dependent employment only — that is, those who are employees. The self-employed, along with several other small categories of workers, are excluded.

In this exercise we will first determine how many workers are excluded from the analysis when the sample is restricted to those in dependent employment. We will then measure the gap in hourly wages for men and women, in each of the three countries in our study.

We will be using a measure of gross hourly wages, which is available in the three datasets we are using. In other datasets, however, it could be that only annual wages were available. In such cases, researchers must take care to account for variations in employment over the year, perhaps by restricting the sample to full-year, full-time workers.

In part I, we have introduced bottom- and top-coding as a technique to deal with extreme values. In this exercise we will also introduce a different technique that deals with extreme values: ‘trimming’ the distribution (i.e. deleting rather than recoding the extreme values). This technique is especially useful when calculating measures that are not defined for non-positive values (such as logarithmic measures). In later exercises we will convert the hourly wages into logs, and thus we need to make sure that the sample that we analyse at this stage is the same that we will keep for our final analysis.

**Activity**

Recode the variable `status1` to create a new variable `depemp` that indicates whether a person is in dependent employment. Using this variable, produce summary statistics reporting the proportion of dependent employment among prime-age male workers and among prime-age female workers, and fill in the following table.
Employment Rates

<table>
<thead>
<tr>
<th>Dataset</th>
<th>Men</th>
<th>Women</th>
</tr>
</thead>
<tbody>
<tr>
<td></td>
<td>Dependent Employment (%)</td>
<td>Non-dependent employment (%)</td>
</tr>
<tr>
<td>US04</td>
<td></td>
<td></td>
</tr>
<tr>
<td>DE04</td>
<td></td>
<td></td>
</tr>
<tr>
<td>GR04</td>
<td></td>
<td></td>
</tr>
</tbody>
</table>

Next, use the LIS hourly wage variable `gross1` to construct a new hourly wage variable `hourwage`, where the bottom and the top of the distribution are corrected as follows:

- at the top, we will carry out the same top-code as used in Part I: hourly wages greater than ten times the national median are set to ten times the national median (make sure you calculate the median and apply the top-code separately for each country);
- at the bottom, we will ‘trim’ the distribution so that observations with negative and zero hourly wages are set to missing.

Using this new hourly wage variable, calculate the gender wage gap for dependent employees in each country. The gender wage gap is defined here as the ratio of the median wages of women to the median wages of men. Use your results to complete the table below.

<table>
<thead>
<tr>
<th></th>
<th>Gender wage gap for dependent employees</th>
</tr>
</thead>
<tbody>
<tr>
<td>US04</td>
<td></td>
</tr>
<tr>
<td>DE04</td>
<td></td>
</tr>
<tr>
<td>GR04</td>
<td></td>
</tr>
</tbody>
</table>

**Question:** Does the percentage of workers not in dependent employment differ substantially across countries? Does it differ between men and women?

**Question:** Which country has the most wage inequality between men and women, among dependent employees?

**Guidelines**

- For the top-coding of hourly wage, apply the same method as in the exercise of part 1. For further discussion of top-coding and bottom-coding, see Part I, exercise 3.
- As before, you can use the `split file` command to produce summary statistics subdivided by country and gender.
Program
define ex24 ().
select if age ge 25 and age le 54.
select if relation ge 1000 and relation le 2200.
weight by ppopwgt.
compute depemp = $sysmis.
if status1 ge 100 and status1 le 120 depemp = 1.
if status1 ge 200 and status1 le 240 depemp = 0.
sort cases by dname.
split file by dname.
aggregate outfile= *
   mode = addvariables
   /break = did
   /medwage= median(gross1).
* frequencies variables = medwage.
*** top and bottomcoding.
compute hourwage = gross1.
if gross1 gt (10*medwage) hourwage = (10*medwage).
if gross1 le 0 hourwage = $sysmis.
temporary.
select if emp = 1.
crosstabs sex by depemp /cells = row.
sort cases by dname sex.
split file by dname sex.
frequencies variables = hourwage /statistics = median /format = notable.
!enddefine .
*match files file = us04p
   /table = us04h
   /keep=did hid own dname pwgt ppopwgt relation partner ageyoch age sex immigr educ educ_c emp status1 ptime1 gross1
   /by hid.
*save outfile = "mydata\us04_yid.sav".
*match files file = de04p
   /table = de04h
   /keep=did hid own dname pwgt ppopwgt relation partner ageyoch age sex immigr educ educ_c emp status1 ptime1 gross1
   /by hid.
*save outfile = "mydata\de04_yid.sav".

*match files file = gr04p
  /table = gr04h
  /keep=did hid own dname pwgt ppopwgt relation partner ageyoch age sex immigr educ educ_c emp status1 ptime1 gross1
  /by hid.

*save outfile = "mydata\gr04_yid.sav".

*add file file = "mydata\us04_yid.sav"
  /file = "mydata\de04_yid.sav"
  /file = "mydata\gr04_yid.sav".

*save outfile = "mydata\stackex_yid.sav".

get file = "mydata\stackex_yid.sav".

ex24.
Results

Employment Rates

<table>
<thead>
<tr>
<th>Dataset</th>
<th>Men</th>
<th>Women</th>
</tr>
</thead>
<tbody>
<tr>
<td></td>
<td>Dependent Employment (%)</td>
<td>Non-dependent employment (%)</td>
</tr>
<tr>
<td>US04</td>
<td>86%</td>
<td>14%</td>
</tr>
<tr>
<td>DE04</td>
<td>88%</td>
<td>12%</td>
</tr>
<tr>
<td>GR04</td>
<td>66%</td>
<td>34%</td>
</tr>
</tbody>
</table>

Gender wage gaps

<table>
<thead>
<tr>
<th></th>
<th>Gender wage gap for dependent employees</th>
</tr>
</thead>
<tbody>
<tr>
<td>US04</td>
<td>0.75</td>
</tr>
<tr>
<td>DE04</td>
<td>0.80</td>
</tr>
<tr>
<td>GR04</td>
<td>0.85</td>
</tr>
</tbody>
</table>

Question: Does the percentage of workers not in dependent employment differ substantially across countries? Does it differ between men and women?

- Greece has a much higher rate of non-dependent employment (which is primarily self-employment). In all the countries, women have somewhat higher rates of dependent employment than men. Keep in mind, therefore, that the results in the subsequent exercises may be somewhat unrepresentative, particularly for Greece, because they exclude a substantial number of workers.

Question: Which country has the most wage inequality between men and women, among dependent employees?

- The United States shows the largest gender wage gap. Among prime age workers, the median hourly wage of women is only 75% that of men.

Comments

- The wage gap calculated here is based on the median, but some researchers calculate an alternative version based on the mean, which will give slightly different results.
5. Hourly wages, education, and country-specific variables

Goal

This exercise continues the analysis of gender wage gaps in hourly wages among those in dependent employment, which we started to program in the previous exercise. In this exercise, we will see how gender disparities in wages differ by educational attainment.

This exercise also demonstrates the use of two different LIS variables measuring educational attainment. One is fully standardised for cross-national compatibility, but contains few categories. The other may contain more information, but has country-specific codes, and thus requires researchers to perform their own standardisation.

The standardised variable is called `educ`, which is based on the International Standard Classification of Education (ISCED). The non-standardised version (from which `educ` is constructed) is `educ_c`. This is one of many attributes for which LIS provides both a standardised and country-specific variable. Any variable ending in `_c` is non-standardised, meaning that it can have different contents in different datasets. It is important to carefully examine the dataset-specific documentation before using such variables.

Activity

Add code to your program to create a table cross-tabulating the variables `educ` and `educ_c` for each country. This will show how the standardised variable was constructed in each case. Be sure to:

- include missing values in your table, so that you can see whether any of the cases in the original education variable could not be allocated to a category in the standardised version;

- remove the value labels from the tabulation of the `educ_c` variable (since the value labels of the `_c` variables are by definition dataset-specific, in a stacked dataset with observations from several LIS datasets, the value labels of those variables will be incorrect, as they can only refer to one specific LIS dataset – usually the last one that was used to construct the stacked data, see further details in the comments section of this exercise).

- Using the hourly wage variable `hourwage` that you created in the last exercise, calculate the gender wage gap by education for each country, and complete the table below. The gender wage gap is
defined as it was in the previous exercise. To obtain the earnings ratio by education, simply calculate the ratio separately for individuals in each of the three categories of the standardised education variable.

**Gender wage gaps by educational attainment**

<table>
<thead>
<tr>
<th></th>
<th>Low education</th>
<th>Medium education</th>
<th>High education</th>
</tr>
</thead>
<tbody>
<tr>
<td>US04</td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>DE04</td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>GR04</td>
<td></td>
<td></td>
<td></td>
</tr>
</tbody>
</table>

**Question:** For each of the three countries, what are the categories in the original dataset that are recoded as “high education” in the standardised education variable?

**Question:** Are there any categories in the original educ_c variable that could not be translated into the standardised form?

**Question:** In general, which educational attainment category shows the greatest earnings inequality between genders? How do the patterns differ by country?

**Guidelines**

- If you are uncertain about the meaning of a code in a categorical variable (like educ_c), consult METIS for variable information and codebook.
- SPSS reminder: to remove the value labels from a variable apply this code when you prepare the stacked datasets:

  Value labels educ_c.

- You can create the medians in two ways, either as in the previous exercise, or by running a second time the aggregate command. The medians can then be added to the file, stored in a new variable name. Both methods work equally well.
- A note on the treatment of missing data in SPSS: unfortunately, system-missing values are not included in the table. There is no subcommand that will enable system-missing values to appear in the crosstab. The closest you could come, is to change the system-missing value to a user-missing value. This could be accomplished with a recode command. But instead for this exercise we recommend to just compare the cross-tabulation against a plain frequency table.
Program

define ex25 ().
select if age ge 25 and age le 54.
select if relation ge 1000 and relation le 2200.
weight by ppopwgt.
sort cases by dname.
split file by dname.
value labels educ_c.
crosstabs educ_c by educ.
frequencies variables = educ_c
aggregate outfile= *
   mode = addvariables
   /break = did
   /medwage= median(gross1).
*** top and bottomcoding.
compute hourwage = gross1.
if gross1 gt (10*medwage) hourwage = (10*medwage).
if gross1 le 0 hourwage = $sysmis.
aggregate outfile= *
   mode = addvariables
   /break = did educ sex
   /mededusex= median(hourwage).
sort cases by dname educ sex.
split file by dname educ sex.
descriptives variables = mededusex.
!enddefine .
*match files file = us04p
   /table = us04h
   /keep=did hid own dname pwgt ppopwgt relation partner ageyoch age sex immigr educ educ_c emp status1 ptime1 gross1
   /by hid.
*save outfile = "mydata\us04_yid.sav".
*match files file = de04p
   /table = de04h
   /keep=did hid own dname pwgt ppopwgt relation partner ageyoch age sex immigr educ educ_c emp status1 ptime1 gross1
   /by hid.
*save outfile = "mydata\de04_yid.sav".

*match files file = gr04p
    /table = gr04h
    /keep=did hid own dname pwgt ppopwg relation partner ageyoch age sex immigr educ educ_c emp status1 ptime1 gross1
    /by hid.

*save outfile = "mydata\gr04_yid.sav".

*add file file = "mydata\us04_yid.sav"
    /file = "mydata\de04_yid.sav"
    /file = "mydata\gr04_yid.sav".

*save outfile = "mydata\stackex_yid.sav".

get file = "mydata\stackex_yid.sav".

ex25.
### Results

#### Gender wage gaps by educational attainment

<table>
<thead>
<tr>
<th></th>
<th>Low education</th>
<th>Medium education</th>
<th>High education</th>
</tr>
</thead>
<tbody>
<tr>
<td>US04</td>
<td>0.71</td>
<td>0.71</td>
<td>0.78</td>
</tr>
<tr>
<td>DE04</td>
<td>0.78</td>
<td>0.83</td>
<td>0.81</td>
</tr>
<tr>
<td>GR04</td>
<td>0.67</td>
<td>0.78</td>
<td>0.96</td>
</tr>
</tbody>
</table>

**Question:** For each of the three countries, what are the categories in the original dataset that are recoded as “high education” in the standardised education variable?

- In the United States, high education combines those with associate degrees, bachelor's degrees, and advanced degrees (masters, professional school, or doctorate).
- In Germany, high education combines those classified as having “higher vocational” (of any kind) or “higher education”.
- In Greece, high education includes those with tertiary graduate level education, postgraduate level education or Ph.D.

**Question:** Are there any categories in the original `educ_c` variable that could not be translated into the standardised form?

- In the United States, all values of `educ_c` receive a value in `educ`. In Germany, a small number of persons categorised as “inadequately, other diploma” or “still in education” are set to missing. In Greece, a small number of persons listed as “still in education” are set to missing.

**Question:** In general, which educational attainment category shows the greatest earnings inequality between genders? How do the patterns differ by country?

- In all three countries, there is a smaller gender wage gap among highly educated workers. This is particularly notable in Greece. In that country, wage inequality is greater among the low-educated than in the United States and Germany, but there is near equality among the highly educated.
Comments

You may have some doubts and questions why we did advise you in the exercise to not show the labels of `educ_c` for the cross-tabulation of `educ` and `educ_c`. As you are aware we did append values for several countries for each variable to get the stacked file. In your stacked file, for standardised variables these values have all the same meaning, as the values and labels are completely standardised. However, it is more complicated for non-standardized values and labels of `_c` variables, as each dataset has its own values and own meaning, as indicated by the labels attached to the data.

Be aware that while appending the data, your programming software will very likely overwrite the label automatically. Thus we do in general advise you to drop the labels from the variables `_c` within your code. You always have the full information on the labels in the codebooks. However, if you prefer to keep the full labels somewhere in the data there are several solutions.

The simple solution is that you tabulate each country separately (see for example exercise 2 of part II) before you generate the stacked version. As a second solution, you can also easily rename the variables `_c` to the specific `_ccyy` of each dataset - this way you will append a separate variable `_ccyy` for each of your datasets, which does have only observations for the specific `ccyy` with the country specific labels attached. Be aware that you then need to tabulate for each `ccyy` separately to get the right percentage of missings!
6. Immigration and wages, understanding harmonisation

Goal
Each of the countries we are examining has a significant immigrant population, and their labor market outcomes are often very different from those of the non-immigrant population. In this exercise we will compare the wages of immigrants and non-immigrant men and women, just as we compared individuals of different educational levels in the last exercise.

LIS provides a variable indicating whether someone is an immigrant, which we will be using in this exercise. However, the choices that go into constructing this variable are complex, because the information available to construct it varies widely from country to country. It is important to understand the assumptions behind variables such as this one, because in some cases researchers may prefer to develop their own standardisation procedures based on their particular needs.

Activity
Using the top and bottom coded hourly wage variable, calculate the gender earnings ratio by immigration status for each country, and complete the table below. The gender earnings ratio is computed just as in the previous exercise, except that you will now subdivide the population into immigrants and non-immigrants, rather than by educational attainment categories.

Gender earnings ratios by immigration status

<table>
<thead>
<tr>
<th></th>
<th>Non-immigrants</th>
<th>Immigrants</th>
</tr>
</thead>
<tbody>
<tr>
<td>US04</td>
<td></td>
<td></td>
</tr>
<tr>
<td>DE04</td>
<td></td>
<td></td>
</tr>
<tr>
<td>GR04</td>
<td></td>
<td></td>
</tr>
</tbody>
</table>

Question: What information is used to construct the immigr variable? If you wanted to determine how the indicator was constructed in a particular dataset, what other variables would you need to look at?

Question: Is gender earnings inequality larger among immigrants or non-immigrants? Does this differ by country?

Guidelines
- The coding required for this exercise is essentially the same as in the previous one. See the guidelines for the previous exercise for details.
- Remember that if you want to comment out the commands from the previous exercise, you can do so by starting the line with a *.
Program

define ex26 ().
select if age ge 25 and age le 54.
select if relation ge 1000 and relation le 2200.
weight by ppopwgt.
aggregate outfile= *
    mode = addvariables
    /break = did
    /medwage= median(gross1).
*** top and bottomcoding.
compute hourwage = gross1.
if gross1 gt (10*medwage) hourwage = (10*medwage).
if gross1 le 0 hourwage = $sysmis.
aggregate outfile= *
    mode = addvariables
    /break = did immigr sex
    /medimmsex= median(hourwage).
sort cases by dname immigr sex.
split file by dname immigr sex.
descriptives variables = medimmsex.
!enddefine.
*match files file = us04p
  /table = us04h
  /keep=did hid own dname pwgt ppopwgt relation partner ageyoch age sex immigr educ educ_c emp status1 ptime1 gross1
  /by hid.
*save outfile = "mydata\us04_yid.sav".
*match files file = de04p
  /table = de04h
  /keep=did hid own dname pwgt ppopwgt relation partner ageyoch age sex immigr educ educ_c emp status1 ptime1 gross1
  /by hid.
*save outfile = "mydata\de04_yid.sav".
*match files file = gr04p
  /table = gr04h
  /keep=did hid own dname pwgt ppopwgt relation partner ageyoch age sex immigr educ educ_c emp status1 ptime1 gross1
  /by hid.
/by hid.
*save outfile = "mydata\gr04_yid.sav".
*add file file = "mydata\us04_yid.sav"
  /file = "mydata\de04_yid.sav"
  /file = "mydata\gr04_yid.sav".
*save outfile = "mydata\stackex_yid.sav".
get file = "mydata\stackex_yid.sav".
ex26.
Results

Gender earnings ratios by immigration status

<table>
<thead>
<tr>
<th></th>
<th>Non-immigrants</th>
<th>Immigrants</th>
</tr>
</thead>
<tbody>
<tr>
<td>US04</td>
<td>0.76</td>
<td>0.82</td>
</tr>
<tr>
<td>DE04</td>
<td>0.80</td>
<td>0.77</td>
</tr>
<tr>
<td>GR04</td>
<td>0.89</td>
<td>0.82</td>
</tr>
</tbody>
</table>

**Question:** What information is used to construct the immig variable? If you wanted to determine how the indicator was constructed in a particular dataset, what other variables would you need to look at?

- As indicated in the Variable Definition of immig using METIS documentation tool, Immigrants are defined by LIS as all persons who have that country as country of usual residence and (in order of priority):
  - whom the data provider defined as immigrants;
  - who self-define themselves as immigrants;
  - who are the citizen/national of another country;
  - who were born in another country.

- The definition of immigrant used in immig may differ substantially from dataset to dataset. The variables that may be used in its construction include citizen, ctrybrth, yrsresid, ethnic_c and immig_c (you can check dataset-specific notes in METIS for information about each dataset).

**Question:** Is gender earnings inequality larger among immigrants or non-immigrants? Does this differ by country?

- In the United States, the gender wage gap is greater among non-immigrants, but in Germany and Greece it is greater among immigrants.
7. **Wage regressions**

**Goal**

We have seen how employment varies by gender and family structure, and how gender wage gaps vary by education and immigration status. In this exercise, we will investigate the impact of all these variables on wages, using a multivariate regression.

Wages are generally not normally distributed. We will therefore apply a logarithmic transformation in order to create an outcome variable that is approximately normal, which is more suitable for regression modeling.

In addition to the variables we have already seen, we will also control for age, which has a strong relationship with earnings. Since the relationship between age and income is not necessarily linear, we will also add a term for age-squared.

**Activity**

Create a new variable, `logwage`, which is the natural logarithm of the `hourwage` variable you created earlier.

Create a variable `agesq`, which is equal to `age` squared.

Create dummy variables based on the `achildcat` and `educ` variables you have used previously. You will need the following variables:

- `youngchild`, equal to 1 if the person has a child under 6 in the household, 0 otherwise.
- `oldchild`, equal to 1 if the person has a child age 6-17 in the household, 0 otherwise.
- `hieduc`, equal to 1 if the person has high educational attainment, 0 otherwise.
- `mededuc`, equal to 1 if the person has medium educational attainment, 0 otherwise.

When including categorical variables in a regression, one category is always left out and used as the reference category. The reference categories in this case are having no non-adult children, and having low educational attainment.

After creating all of the variables, run regression models predicting `logwage`. You should run the regressions separately in each country, and within each country you should run a separate model for men and women. Use the following model:
logwage = f (age agesq mededuc hieduc immigr partner youngchild oldchild ptime1 homeowner)

Produce a table of the six resulting models, with coefficients, standard errors, sample sizes, and r-squared values.

It is recommended that you create a new sub-routine to run the regressions, which can be called from your main program.

**Question:** Who receives a higher wage premium from being highly educated, men or women?

**Question:** When controlling for other individual characteristics, what is the relationship between immigrant status and wages?

**Question:** When controlling for other individual characteristics, do women with young children make more or less than women without children?

**Guidelines**

- Linear regression in SPSS is done with the `regression` command:

  ```
  regression /regwt=<weight variable>
  /dependent=<dependent variable>
  /method=enter <independent variables>.
  ```
Program

define ex27 ().
select if age ge 25 and age le 54.
select if relation ge 1000 and relation le 2200.
weight by ppopwgt.
if own ge 100 and own le 199 homeowner = 1.
if own ge 200 and own le 299 homeowner = 0.
recode ageyoch (0 thru 5=1) (6 thru 17=2) (ELSE=0) into achildcat.
add value labels achildcat
   0 "no children under 18"
   1 "under 6 years"
   2 "6-17 years".
compute depemp = $sysmis.
if status1 ge 100 and status1 le 120 depemp = 1.
if status1 ge 200 and status1 le 240 depemp = 0.
*** top and bottomcoding.
compute hourwage = gross1.
if gross1 le 0 hourwage = $sysmis.
aggregate outfile= *
   mode = addvariables
   /break = did
   /medwage= median(hourwage).
if gross1 gt (10*medwage) hourwage = (10*medwage).
compute logwage = ln(hourwage).
compute agesq = age*age.
if achildcat eq 1 youngchild = 1.
if achildcat eq 0 or achildcat eq 2 youngchild = 0.
if achildcat eq 2 oldchild = 1.
if achildcat eq 0 or achildcat eq 1 oldchild = 0.
if educ eq 2 mededuc = 1.
if educ eq 1 or educ eq 3 mededuc = 0.
if educ eq 3 hieduc = 1.
if educ eq 1 or educ eq 2 hieduc = 0.
*aggregate outfile= *
   mode = addvariables
   /break = did immigr sex
/medimmsex= median(hourwage).

sort cases by dname sex.
split file by dname sex.
weight off.
regression /regwgt=ppopwgt
  /dependent=logwage
  /method=enter age agesq mededuc hieduc immigr partner youngchild oldchild ptime1 homeowner.
!enddefine .
*match files file = us04p
  /table = us04h
  /keep=did hid own dname pwgt ppopwgt relation partner ageyoch age sex immigr educ educ_c emp status1 ptime1 gross1
  /by hid.
*save outfile = "mydata\us04_yid.sav".
*match files file = de04p
  /table = de04h
  /keep=did hid own dname pwgt ppopwgt relation partner ageyoch age sex immigr educ educ_c emp status1 ptime1 gross1
  /by hid.
*save outfile = "mydata\de04_yid.sav".
*match files file = gr04p
  /table = gr04h
  /keep=did hid own dname pwgt ppopwgt relation partner ageyoch age sex immigr educ educ_c emp status1 ptime1 gross1
  /by hid.
*save outfile = "mydata\gr04_yid.sav".
*add file file = "mydata\us04_yid.sav"
  /file = "mydata\de04_yid.sav"
  /file = "mydata\gr04_yid.sav".
*save outfile = "mydata\stackex_yid.sav".
get file = "mydata\stackex_yid.sav".
ex27.
### Results

#### Men

<table>
<thead>
<tr>
<th></th>
<th>us04</th>
<th>de04</th>
<th>gr04</th>
</tr>
</thead>
<tbody>
<tr>
<td>age</td>
<td>0.050***</td>
<td>0.055***</td>
<td>0.033</td>
</tr>
<tr>
<td></td>
<td>(0.0056)</td>
<td>(0.015)</td>
<td>(0.017)</td>
</tr>
<tr>
<td>agesq</td>
<td>-0.00049***</td>
<td>-0.00061**</td>
<td>-0.00023</td>
</tr>
<tr>
<td></td>
<td>(0.000070)</td>
<td>(0.00019)</td>
<td>(0.00021)</td>
</tr>
<tr>
<td>mededuc</td>
<td>0.28***</td>
<td>0.012</td>
<td>0.17***</td>
</tr>
<tr>
<td></td>
<td>(0.015)</td>
<td>(0.068)</td>
<td>(0.028)</td>
</tr>
<tr>
<td>hieduc</td>
<td>0.68***</td>
<td>0.40***</td>
<td>0.43***</td>
</tr>
<tr>
<td></td>
<td>(0.016)</td>
<td>(0.070)</td>
<td>(0.037)</td>
</tr>
<tr>
<td>immigr</td>
<td>-0.058***</td>
<td>-0.067*</td>
<td>-0.25***</td>
</tr>
<tr>
<td></td>
<td>(0.012)</td>
<td>(0.029)</td>
<td>(0.044)</td>
</tr>
<tr>
<td>partner</td>
<td>0.041***</td>
<td>0.024</td>
<td>-0.014</td>
</tr>
<tr>
<td></td>
<td>(0.013)</td>
<td>(0.036)</td>
<td>(0.042)</td>
</tr>
<tr>
<td>youngchild</td>
<td>0.062***</td>
<td>0.070*</td>
<td>0.12***</td>
</tr>
<tr>
<td></td>
<td>(0.012)</td>
<td>(0.034)</td>
<td>(0.034)</td>
</tr>
<tr>
<td>oldchild</td>
<td>0.069***</td>
<td>0.047</td>
<td>0.11***</td>
</tr>
<tr>
<td></td>
<td>(0.011)</td>
<td>(0.026)</td>
<td>(0.032)</td>
</tr>
<tr>
<td>ptime1</td>
<td>-0.30***</td>
<td>-0.30**</td>
<td>0.33***</td>
</tr>
<tr>
<td></td>
<td>(0.031)</td>
<td>(0.096)</td>
<td>(0.084)</td>
</tr>
<tr>
<td>homeowner</td>
<td>0.24***</td>
<td>0.11***</td>
<td>0.035</td>
</tr>
<tr>
<td></td>
<td>(0.012)</td>
<td>(0.024)</td>
<td>(0.026)</td>
</tr>
<tr>
<td>_cons</td>
<td>1.11***</td>
<td>1.21***</td>
<td>0.81*</td>
</tr>
<tr>
<td></td>
<td>(0.11)</td>
<td>(0.31)</td>
<td>(0.32)</td>
</tr>
<tr>
<td>N</td>
<td>29986</td>
<td>3652</td>
<td>1175</td>
</tr>
<tr>
<td>R-sq</td>
<td>0.223</td>
<td>0.167</td>
<td>0.346</td>
</tr>
</tbody>
</table>

Standard errors in parentheses
* p<0.05, ** p<0.01, *** p<0.001
### Women

<table>
<thead>
<tr>
<th></th>
<th>us04</th>
<th>de04</th>
<th>gr04</th>
</tr>
</thead>
<tbody>
<tr>
<td>age</td>
<td>0.042***</td>
<td>0.051**</td>
<td>0.071***</td>
</tr>
<tr>
<td></td>
<td>(0.0059)</td>
<td>(0.017)</td>
<td>(0.021)</td>
</tr>
<tr>
<td>agesq</td>
<td>-0.00044***</td>
<td>-0.00057**</td>
<td>-0.00065*</td>
</tr>
<tr>
<td></td>
<td>(0.000074)</td>
<td>(0.00021)</td>
<td>(0.00027)</td>
</tr>
<tr>
<td>mededuc</td>
<td>0.32***</td>
<td>0.083*</td>
<td>0.35***</td>
</tr>
<tr>
<td></td>
<td>(0.018)</td>
<td>(0.039)</td>
<td>(0.040)</td>
</tr>
<tr>
<td>hieduc</td>
<td>0.75***</td>
<td>0.38***</td>
<td>0.76***</td>
</tr>
<tr>
<td></td>
<td>(0.018)</td>
<td>(0.042)</td>
<td>(0.043)</td>
</tr>
<tr>
<td>immigr</td>
<td>-0.018*</td>
<td>-0.070*</td>
<td>-0.26***</td>
</tr>
<tr>
<td></td>
<td>(0.013)</td>
<td>(0.029)</td>
<td>(0.051)</td>
</tr>
<tr>
<td>partner</td>
<td>-0.012</td>
<td>-0.016</td>
<td>-0.067</td>
</tr>
<tr>
<td></td>
<td>(0.011)</td>
<td>(0.032)</td>
<td>(0.040)</td>
</tr>
<tr>
<td>youngchild</td>
<td>0.039**</td>
<td>-0.00079</td>
<td>0.058</td>
</tr>
<tr>
<td></td>
<td>(0.013)</td>
<td>(0.045)</td>
<td>(0.042)</td>
</tr>
<tr>
<td>oldchild</td>
<td>-0.048***</td>
<td>-0.12***</td>
<td>0.026</td>
</tr>
<tr>
<td></td>
<td>(0.010)</td>
<td>(0.031)</td>
<td>(0.036)</td>
</tr>
<tr>
<td>ptime1</td>
<td>-0.17***</td>
<td>-0.10***</td>
<td>0.23***</td>
</tr>
<tr>
<td></td>
<td>(0.012)</td>
<td>(0.028)</td>
<td>(0.041)</td>
</tr>
<tr>
<td>homeowner</td>
<td>0.19***</td>
<td>0.072**</td>
<td>-0.0083</td>
</tr>
<tr>
<td></td>
<td>(0.011)</td>
<td>(0.026)</td>
<td>(0.032)</td>
</tr>
<tr>
<td>_cons</td>
<td>1.11***</td>
<td>1.25***</td>
<td>-0.26</td>
</tr>
<tr>
<td></td>
<td>(0.11)</td>
<td>(0.33)</td>
<td>(0.39)</td>
</tr>
</tbody>
</table>

N: 28597 3575 966

R-sq: 0.190 0.119 0.461

Standard errors in parentheses
* p<0.05, ** p<0.01, *** p<0.001
**Question:** Who receives a higher wage premium from being highly educated, men or women?

- In the US and Greece, the coefficient for high education is higher for women, indicating a larger wage premium from having high educational attainment, while in Germany the premium is slightly higher for men.

**Question:** When controlling for other individual characteristics, what is the relationship between immigrant status and wages?

- The association between immigrant status and wages is negative in all three countries. In general, the negative association appears stronger in Germany than in the United States, and stronger in Greece than Germany.

**Question:** When controlling for other individual characteristics, do women with young children make more or less than women without children?

- In general, women with young children have higher wages than women without children in the United States and Greece, but there is no association in Germany. Higher wages for women with young children could be due to a selection effect, where mothers of young children are more likely to enter the labor market if they have higher earning power.

**Comments**

- As we have seen, employment rates, particularly among women, vary substantially across countries. The wage regressions shown here do not account for this differential selection into employment. For this reason, many studies of wages apply a technique such as a Heckman correction, which attempts to correct for this selection bias.
8. Pooled regressions and normalised weights

Goal

In the previous exercise, we ran parallel, separate regressions for each country. In this exercise, we see an alternative approach, in which all countries are pooled together in a single model. We will continue to use classical OLS regression, but the approach shown here can easily be extended for more complex multilevel estimation approaches.

The income variables in these datasets use different currencies. To compare them, we need to convert them to a common scale. We will apply *Purchasing Power Parity* (PPP) deflators, which are intended to ensure that equal quantities of income correspond to equivalent purchasing power across currencies and national economies. The PPP deflators provided here are taken from the Penn World Tables, which are a commonly used source.

Up to this point, we have been using the weight variable `ppopwgt`, which inflates to the total population. If we use this weight in a pooled regression, every household will receive equal weight. However, this would mean that Greece — which has a much smaller population than the United States or Germany — will not have much influence on the results. In order to give each country equal weight in the model, we will use the alternative normalized weight variable `pwgt`, which always sums to 10,000 within each dataset.

Activity

Create a new logged hourly wage variable, `logwage_ppp` by dividing `hourwage` by the following deflators ([http://www.lisdatacenter.org/data-access/web-tabulator/methods/ppp/](http://www.lisdatacenter.org/data-access/web-tabulator/methods/ppp/)) and then taking the natural logarithm:

<table>
<thead>
<tr>
<th>dataset</th>
<th>PPP deflator</th>
</tr>
</thead>
<tbody>
<tr>
<td>US04</td>
<td>1</td>
</tr>
<tr>
<td>DE04</td>
<td>0.74</td>
</tr>
<tr>
<td>GR04</td>
<td>0.62</td>
</tr>
</tbody>
</table>

Create dummy variables for `germany` (=1 if the case is from German dataset, otherwise 0) and `greece` (=1 if the case is from the Greek dataset, otherwise 0). The United States will be the reference category.

Estimate the following model, for men and women separately:

\[
\text{logwage} = f (\text{age} \text{agesq} \text{mededuc} \text{hieduc} \text{immigr} \text{partner} \text{youngchild} \text{oldchild} \text{ptime1} \text{homeowner} \text{germany} \text{greece})
\]
Make sure to use *normalised*, not inflated weights.

Produce a table of the two resulting models, with coefficients, standard errors, sample sizes, and r-squared values.

**Question:** How can you interpret the meaning of the coefficients for the Germany and Greece dummy variables?

**Question:** In this pooled model, which carries a higher wage penalty: being an immigrant, or working part time?

**Guidelines**

- Run your regressions as you did in the previous exercise. This time, make sure to
  - use *normalized* (pwgt) not inflated weights (ppopwgt)
  - sort your dataset on which your run the regression only by *sex* and not by *dname sex* since all countries are pooled together in a single model.
Program

define ex28 ().
select if age ge 25 and age le 54.
select if relation ge 1000 and relation le 2200.
weight by ppopwgt.
if own ge 100 and own le 199 homeowner = 1.
if own ge 200 and own le 299 homeowner = 0.
recode ageyoch (0 thru 5=1) (6 thru 17=2) (ELSE=0) into achildcat.
add value labels achildcat
  0 "no children under 18"
  1 "under 6 years"
  2 "6-17 years".
compute depemp = $sysmis.
if status1 ge 100 and status1 le 120 depemp = 1.
if status1 ge 200 and status1 le 240 depemp = 0.
*** top and bottomcoding.
compute hourwage = gross1.
if gross1 le 0 hourwage = $sysmis.
aggregate outfile= *
  mode = addvariables
  /break = did
  /medwage= median(hourwage).
if gross1 gt (10*medwage) hourwage = (10*medwage).
compute logwage = ln(hourwage).
compute agesq = age*age.
if achildcat eq 1 youngchild = 1.
if achildcat eq 0 or achildcat eq 2 youngchild = 0.
if achildcat eq 2 oldchild = 1.
if achildcat eq 0 or achildcat eq 1 oldchild = 0.
if educ eq 2 mededuc = 1.
if educ eq 1 or educ eq 3 mededuc = 0.
if educ eq 3 hieduc = 1.
if educ eq 1 or educ eq 2 hieduc = 0.
if dname eq "de04" ppp=0.74.
if dname eq "gr04" ppp=0.62.
if dname eq "us04" ppp=1.
compute hourwage_ppp = hourwage/ppp.
compute logwage_ppp = ln(hourwage_ppp).
compute germany=0.
if dname eq "de04" germany=1.
compute greece=0.
if dname eq "gr04" greece =1.
*aggregate outfile= *
   mode = addvariables
   /break = did immigr sex
   /medimmsex= median(hourwage).
sort cases by sex.
split file by sex.
weight off.
regression /regwgt=pwgt
   /dependent=logwage_ppp
   /method=enter age agesq mededuc hieduc immigr partner youngchild oldchild ptime1 homeowner germany greece.
!enddefine .
*match files file = us04p
  /table = us04h
  /keep=did hid own dname pwgt ppopwgt relation partner ageyoch age sex immigr educ educ_c emp status1 ptime1 gross1
  /by hid.
*save outfile = "mydata\us04_yid.sav".
*match files file = de04p
  /table = de04h
  /keep=did hid own dname pwgt ppopwgt relation partner ageyoch age sex immigr educ educ_c emp status1 ptime1 gross1
  /by hid.
*save outfile = "mydata\de04_yid.sav".
*match files file = gr04p
  /table = gr04h
  /keep=did hid own dname pwgt ppopwgt relation partner ageyoch age sex immigr educ educ_c emp status1 ptime1 gross1
  /by hid.
*save outfile = "mydata\gr04_yid.sav".
*add file file = "mydata\us04_yid.sav"
   /file = "mydata\de04_yid.sav"
   /file = "mydata\gr04_yid.sav".
*save outfile = "mydata\stackex_yid.sav".
get file = "mydata\stackex_yid.sav".
ex28.
### Results

<table>
<thead>
<tr>
<th></th>
<th>Men</th>
<th>Women</th>
</tr>
</thead>
<tbody>
<tr>
<td>age</td>
<td>0.052***</td>
<td>0.057***</td>
</tr>
<tr>
<td></td>
<td>(0.0070)</td>
<td>(0.0079)</td>
</tr>
<tr>
<td>agesq</td>
<td>-0.00052***</td>
<td>-0.00060***</td>
</tr>
<tr>
<td></td>
<td>(0.000089)</td>
<td>(0.000099)</td>
</tr>
<tr>
<td>mededuc</td>
<td>0.19***</td>
<td>0.24***</td>
</tr>
<tr>
<td></td>
<td>(0.020)</td>
<td>(0.022)</td>
</tr>
<tr>
<td>hieduc</td>
<td>0.53***</td>
<td>0.65***</td>
</tr>
<tr>
<td></td>
<td>(0.022)</td>
<td>(0.022)</td>
</tr>
<tr>
<td>immigr</td>
<td>-0.12***</td>
<td>-0.077***</td>
</tr>
<tr>
<td></td>
<td>(0.015)</td>
<td>(0.017)</td>
</tr>
<tr>
<td>partner</td>
<td>0.037*</td>
<td>-0.014</td>
</tr>
<tr>
<td></td>
<td>(0.018)</td>
<td>(0.016)</td>
</tr>
<tr>
<td>youngchild</td>
<td>0.076***</td>
<td>0.016</td>
</tr>
<tr>
<td></td>
<td>(0.016)</td>
<td>(0.019)</td>
</tr>
<tr>
<td>oldchild</td>
<td>0.070***</td>
<td>-0.067**</td>
</tr>
<tr>
<td></td>
<td>(0.013)</td>
<td>(0.015)</td>
</tr>
<tr>
<td>ptime1</td>
<td>-0.13**</td>
<td>-0.057***</td>
</tr>
<tr>
<td></td>
<td>(0.047)</td>
<td>(0.016)</td>
</tr>
<tr>
<td>homeowner</td>
<td>0.14***</td>
<td>0.086***</td>
</tr>
<tr>
<td></td>
<td>(0.013)</td>
<td>(0.014)</td>
</tr>
<tr>
<td>greece</td>
<td>-0.36***</td>
<td>-0.20***</td>
</tr>
<tr>
<td></td>
<td>(0.014)</td>
<td>(0.017)</td>
</tr>
<tr>
<td>_cons</td>
<td>1.26***</td>
<td>0.95***</td>
</tr>
<tr>
<td></td>
<td>(0.14)</td>
<td>(0.15)</td>
</tr>
</tbody>
</table>

| N        | 34813        | 33138         |
| R-sq     | 0.292        | 0.228         |

Standard errors in parentheses
* p<0.05, ** p<0.01, *** p<0.001
**Question:** How can you interpret the meaning of the coefficients for the Germany and Greece dummy variables?

- These coefficients represent the overall national level of PPP-adjusted wages, when controlling for the other variables. The small positive value for Germany reflects the fact that Germany is a slightly richer country than the United States, while the negative value for Greece reflects the fact that Greece is poorer.

**Question:** In this pooled model, which carries a higher wage penalty: being an immigrant, or working part time?

- For women, the wage penalty for working part time is smaller than that for being an immigrant. The reverse is true for men, although it should be noted that part time work is very uncommon among prime age men.